The Chain Rule

```
>
> with(linalg):
Warning, new definition for norm
Warning, new definition for trace
```

If f is an unassigned letter, Maple treats it as an arbitrary constant (possible complex valued). If f is an unassigned letter, the notation f(x,y,z) is an arbitrary value that depends on x, y, and z. In this way we can regard f as a function, and f(x,y,z) (an expression) which represents the value of the function at x, y, and z. To represent an arbitrary vector valued function we need to impose a little more structure.

```
> f:=map(unapply,vector([f1(x,y,z),f2(x,y,z),f3(x,y,z)]),x,y,z);
```

$$f := [f1, f2, f3]$$

```
> f(x,y,z);
```

$$[\mathrm{f1}(x, y, z), \mathrm{f2}(x, y, z), \mathrm{f3}(x, y, z)]$$

The choice of letters is of no concern in our definition of f, so we could regard f as a function of u, v, and w.

```
> f(u,v,w);
```

$$[\mathrm{f1}(u, v, w), \mathrm{f2}(u, v, w), \mathrm{f3}(u, v, w)]$$

I do not wish to pursue this matter any further, but at least now you have some ideas on how to create an arbitrary vector valued function, if you ever need to create one.

Define g:R^2 -> R^3 and f:R^3 -> R^3 by

```
> g:=map(unapply,vector([x^2-y^2,5*x*y-8,4*x-3*y]),x,y);
  f:=map(unapply,vector([u*v*w,2*u+3*v+4*w,
  8*exp(2*u+v^2-2*w)]),u,v,w);
```

$$g := [(x, y) \to x^2 - y^2, (x, y) \to 5\,x\,y - 8, (x, y) \to 4\,x - 3\,y]$$

$$f := [(u, v, w) \to u\,v\,w, (u, v, w) \to 2\,u + 3\,v + 4\,w, (u, v, w) \to 8\,\mathbf{e}^{(2\,u + v^2 - 2\,w)}]$$

```
> u1:=g[1](x,y); v1:=g[2](x,y);w1:=g[3](x,y);
```

$$u1 := x^2 - y^2$$

$$v1 := 5\,x\,y - 8$$

$$w1 := 4\,x - 3\,y$$

```
> fg:=f(u1,v1,w1);
```

$$fg := [(x^2 - y^2)\,(5\,x\,y - 8)\,(4\,x - 3\,y), 2\,x^2 - 2\,y^2 + 15\,x\,y - 24 + 16\,x - 12\,y,$$
$$8\,\mathbf{e}^{(2\,x^2 - 2\,y^2 + (5\,x\,y - 8)^2 - 8\,x + 6\,y)}]$$

We compute the derivative of fg at (x,y)=(2,1) and compare it to the answer we get by using the Chain Rule.

```
> Dfg:=jacobian(fg,[x,y]);
```

$$Dfg :=$$
$$[2\,x\,(5\,x\,y - 8)\,(4\,x - 3\,y) + 5\,(x^2 - y^2)\,y\,(4\,x - 3\,y) + 4\,(x^2 - y^2)\,(5\,x\,y - 8),$$
$$-2\,y\,(5\,x\,y - 8)\,(4\,x - 3\,y) + 5\,(x^2 - y^2)\,x\,(4\,x - 3\,y) - 3\,(x^2 - y^2)\,(5\,x\,y - 8)]$$
$$[4\,x + 15\,y + 16, -4\,y + 15\,x - 12]$$
$$[8\,(4\,x + 10\,(5\,x\,y - 8)\,y - 8)\,\mathbf{e}^{(2\,x^2 - 2\,y^2 + (5\,x\,y - 8)^2 - 8\,x + 6\,y)},$$
$$8\,(-4\,y + 10\,(5\,x\,y - 8)\,x + 6)\,\mathbf{e}^{(2\,x^2 - 2\,y^2 + (5\,x\,y - 8)^2 - 8\,x + 6\,y)}]$$

```
> Dfg0:=subs(x=2,y=1,evalm(Dfg));
```

$$Dfg0 := \begin{bmatrix} 139 & 112 \\ 39 & 14 \\ 160\ \mathbf{e}^0 & 336\ \mathbf{e}^0 \end{bmatrix}$$

```
> Dfg0:=simplify(Dfg0);
```

$$Dfg0 := \begin{bmatrix} 139 & 112 \\ 39 & 14 \\ 160 & 336 \end{bmatrix}$$

Using the Chain Rule.

```
> Df:=jacobian(f(u,v,w),[u,v,w]);
```

$$Df := \begin{bmatrix} \frac{v\,w}{2} & \frac{u\,w}{3} & \frac{u\,v}{4} \\ 16\ \mathbf{e}^{(2\,u+v^2-2\,w)} & 16\,v\ \mathbf{e}^{(2\,u+v^2-2\,w)} & -16\ \mathbf{e}^{(2\,u+v^2-2\,w)} \end{bmatrix}$$

```
> g(2,1);
```

$$[3, 2, 5]$$

```
> Df0:=subs(u=3,v=2,w=5,evalm(Df));
```

$$Df0 := \begin{bmatrix} 10 & 15 & 6 \\ 2 & 3 & 4 \\ 16\ \mathbf{e}^0 & 32\ \mathbf{e}^0 & -16\ \mathbf{e}^0 \end{bmatrix}$$

```
> Df0:=simplify(Df0);
```

$$Df0 := \begin{bmatrix} 10 & 15 & 6 \\ 2 & 3 & 4 \\ 16 & 32 & -16 \end{bmatrix}$$

```
> Dg:=jacobian(g(x,y),[x,y]);
```

$$Dg := \begin{bmatrix} 2\,x & -2\,y \\ 5\,y & 5\,x \\ 4 & -3 \end{bmatrix}$$

```
> Dg0:=subs(x=2,y=1,evalm(Dg));
```

$$Dg0 := \begin{bmatrix} 4 & -2 \\ 5 & 10 \\ 4 & -3 \end{bmatrix}$$

```
> multiply(Df0,Dg0);
```

$$eq := Dfg0 = \begin{bmatrix} 139 & 112 \\ 39 & 14 \\ 160 & 336 \end{bmatrix}$$

```
> evalm(Dfg0);
```

$$\begin{bmatrix} 139 & 112 \\ 39 & 14 \\ 160 & 336 \end{bmatrix}$$

Yea!!!!!!!!!!!!!!!!!!!!!!!!!!!!

Notice that Maple can use the Chain Rule in a more direct way, without forming matrix derivatives.
We did problem 6 in class on Tuesday. With Maple:

```
> restart:
> z:=f(x+y,x-y);
```

$$z := f(x+y, x-y)$$

```
> zx:=diff(z,x);zy:=diff(z,y);
```

$$zx := D_1(f)(x+y, x-y) + D_2(f)(x+y, x-y)$$

$$zy := D_1(f)(x+y, x-y) - D_2(f)(x+y, x-y)$$

```
> zxzy:=zx*zy;
```

$$zxzy := (D_1(f)(x+y, x-y) + D_2(f)(x+y, x-y))\,(D_1(f)(x+y, x-y) - D_2(f)(x+y, x-y))$$

```
> 'zxzy'=expand(zxzy);
```

$$zxzy = D_1(f)(x+y, x-y)^2 - D_2(f)(x+y, x-y)^2$$

Notice the role of the single quotes. They postpone evaluation. Try the same line without the single quotes.

```
>
```

Implicit differentiation. If $\mathbf{f}:R^{(n+p)}$ -> R^n , say f=(f_1, f_2 , . . . f_n) , and x=(x_1, x_2, . . . x_{n+p}) then an equation of the form **f(x)**=**0**=(0, 0,0, . . . ,0), the 0-vector in R^n represents a system of n equations in n+p unknowns x_1, x_2 ,. . . x_{n+p} . Intuitively we think that we treat n of the x's as unknowns and think of this as a system of n equations in n unknowns, we should be able so solve for the unknowns in terms of the remaining x's. It would be hard (usually impossible) to actually find the formulas for the solutions for all but the most trivial systems. Nevertheless, we can still find the partial derivatives of the solutions by implicit differentiation.

Suppose we look at the system f1(x,y,z)=0, f2(x,y,z)=0. Intuitively we think that if we treat x, and y as unknowns we should be able to solve the system to get x=g(z), and y=h(z). We compute $\frac{dx}{dz}$ and $\frac{dy}{dz}$.

```
> restart;
> eq1:=f1(g(z),h(z),z)=0;
```

$$eq1 := f1(g(z), h(z), z) = 0$$

```
> deq1:=diff(eq1,z);
```

$deq1 :=$

$$D_1(f1)(g(z), h(z), z)\left(\frac{\partial}{\partial z} g(z)\right) + D_2(f1)(g(z), h(z), z)\left(\frac{\partial}{\partial z} h(z)\right) + D_3(f1)(g(z), h(z), z) = 0$$

I won't finish the problem, but now that we have a derivative, it helps a great deal to get rid of this clumsy notation for a derivative. Watch:

```
> deq11:=subs(diff(g(z),z)=dgz,diff(h(z),z)=dhz,deq1);
```

$$deq11 := D_1(f1)(g(z), h(z), z)\, dgz + D_2(f1)(g(z), h(z), z)\, dhz + D_3(f1)(g(z), h(z), z) = 0$$

This technique can be used on abstract systems of equations, or they could be used in the same way on actual systems obtained by specifying formulas for f1 and f2. Try finishing this problem before you go on to the assigned Maple problems.